Electronic Supplementary Material (ESI) for ChemComm.

Lighting up spin systems: enhancing characteristic ^{1}H signal patterns of fluorinated molecules

Electronic Supporting Information

*Marshall J. Smith,[a] Jack E. Bramham,[a] Mathias Nilsson,[a] Gareth A. Morris,[a] Laura Castañar,[*a,b] and Alexander P. Golovanov[*a]*

[a] Department of Chemistry, The University of Manchester, Oxford Road, Manchester, M13 9PL, UK.
[b] Department of Organic Chemistry, Faculty of Chemical Science, Complutense University of Madrid, Ciudad Universitaria s/n, 28040 Madrid, Spain.
[*] Correspondence to a.golovanov@manchester.ac.uk and lcastana@ucm.es

Table of Contents

A. Experimental section

Unless otherwise stated, all data were acquired on a Bruker 500 MHz Avance III spectrometer using a 5 mm QCI-F cryoprobe equipped with a z-gradient coil with a maximum nominal gradient strength of 0.53 T m^{-1}. The model system used in this study contained 1 mM 6-fluoroindole (6FI, Fluorochem) and 0.2 mM riboflavin 5'-monophosphate sodium salt hydrate (FMN, Sigma-Aldrich) in D_2O, as described previously.[1] No attempts were made to degas samples or remove oxygen.

The pulse sequence used in this work, Light-FESTA, is shown in Figure S1. Narrow black and grey rectangles represent hard 90° and 180° pulses, respectively. The durations of the hard 90° pulses were set to 10.45 μs and 12.38 μs for ^{1}H and ^{19}F, respectively. The phases of the pulses are shown in Table S1. For selective observation of a particular fluorinated molecule in a mixture of multiple fluorinated species, the initial hard ^{19}F 90° pulse should be replaced by a selective 90° excitation pulse to select a single resonance of choice.[2] If ^{19}F-^{19}F couplings are present in the system, the hard ^{19}F 180° pulses should be replaced by selective 180° pulses. In this work, a single monofluorinated species was used so all ^{19}F pulses were hard. In all cases the signals of species not containing fluorine, or containing fluorines that are not excited by the selective pulse(s), are suppressed, making this experiment suitable for targeted observation of probe molecules in multicomponent mixtures. The shaped ^{1}H pulses indicate selective 180° refocusing pulses, which were 23 ms (80 Hz bandwidth) Rsnob pulses. The use of selective ^{1}H 180° pulses avoids J_{HH} modulation, allowing for higher sensitivity in the FESTA spectra (Figure S7). The delays Δ_1 and Δ_2 were set to $1/(4J_{HF}\eta_H)$ and $1/(4J_{HF}\eta_F)$, where η_H is the number of selected protons coupled to the selected fluorine and η_F is the number of equivalent ^{19}F nuclei selected. In this work, the Δ delays were optimised for each of the selected signals using the following values: J_{H4F} 5.5 Hz, J_{H5F} 9.5 Hz and J_{H7F} 10.2 Hz. Isotropic mixing was achieved with a DIPSI-2[3] element which, unless otherwise stated, had a duration of 200 ms. The trapezoids before and after the DIPSI-2 element represent 180° chirp pulses used to suppress zero-quantum coherences;[4] their durations were set to 15 and 20 ms, respectively. The light grey trapezoids indicate field gradient pulses; G_1, G_2, G_4 and G_5 are used to enforce the coherence transfer pathway and had durations of 1 ms. The ratio of the amplitudes of G_2 and G_4 was set to 1:0.9407. G_3 and G_7 are homospoil gradient pulses with a duration of 1 ms. The grey rectangles labelled G_6 and G_8 indicate gradients applied during the chirp pulses for zero-quantum suppression (ZQS). All gradient pulses were followed by a recovery delay of 200 μs. All ^{1}H NMR spectra were acquired with a 10 ppm spectral width and 16k complex points. The sample was illuminated using an LED-based NMRtorch device with 460 nm peak emission and 3 W power consumption.[1] Illumination was applied during the recovery delay, d1, for 2 s.



Figure S1. Light-FESTA pulse sequence. The narrow black and wide grey rectangles represent hard 90° and 180° pulses, respectively. The shaped pulse represents a selective 180° radiofrequency pulse applied to a ^{1}H resonance coupled to ^{19}F. The trapezoids on either side of the isotropic mixing element (DIPSI-2) represent zero-quantum suppression elements. Illumination (*hν*) is applied during relaxation delay, d1. If a specific molecule needs to be observed in a mixture of multiple fluorinated molecules then the first ^{19}F hard pulse should be replaced with an appropriate frequency-selective shaped pulse, so that only a single fluorine signal is excited.

Table S1. Phase cycle for the Light-FESTA pulse sequence shown in Figure S1.

ϕ_1	$\phi_{2,3,5,6,7,10,11,12}$	$\phi_{4,8}$	ϕ_{rec}
x, -x, -x, x	x	y	x, -x, -x, x

B. Scheme of work

The steps below outline the suggested scheme of work for obtaining optimal results using the method described in the manuscript. Steps that are not strictly required but can be used to provide further information are marked as optional.

1. Acquire direct ^{1}H and ^{1}H{^{19}F} NMR spectra without illumination.
2. Acquire direct ^{19}F and ^{19}F{^{1}H} (optional) NMR spectra without illumination.
3. Acquire the direct ^{19}F NMR spectrum with illumination to ensure a signal enhancement is observed.
4. Select the fluorine signal of interest and use the data from steps 1 and 2 to determine the number of protons coupled to the selected fluorine and the heteronuclear scalar coupling (J_{HF}) between them.
5. (Optional) In cases where J_{HF} values and ^{19}F-^{1}H coupled partners cannot be determined from the ^{1}H and ^{19}F spectra (step 4), acquire an array of Selective Reverse INEPT (SRI) spectra with illumination (see pulse sequence shown below),[2, 5] exciting the fluorine signal of interest and with different Δ_1 values to obtain information on the ^{1}H signals that are coupled to the selected ^{19}F nucleus and the heteronuclear scalar coupling (J_{HF}) between them.



6. (Optional) Acquire the doubly selective Selective Reverse INEPT – Selective Modulated Echo(SRI-SME) spectrum with illumination (see pulse sequence shown below)[2, 6] with the optimal parameters determined in step 4 or step 5 to confirm the efficiency of the polarisation transfer.



7. Acquire the Light-FESTA spectrum using the pulse sequence shown in Figure S1, to obtain a simplified ^{1}H subspectrum that only contains ^{1}H signals that are within the same spin system as the selected ^{1}H-^{19}F coupled pair.
8. (Optional) Perform Light-FESTA experiments for each of the ^{19}F-^{1}H coupled partners identified in step 4 or 5.

C. Experimental data

1. Enlargement of ^{1}H NMR spectra shown in Figure 1 of the manuscript



Figure S2. 500 MHz ^{1}H NMR spectra of 1 mM 6FI with 0.2 mM riboflavin 5'-monophosphate sodium salt hydrate in D_2O, (a) in the dark state (blue) and (b) after 2 s illumination (red). The illuminated direct ^{1}H photo-CIDNP spectrum (b) shows severe distortion of the multiplets and relative signal intensities, rendering the spectral fingerprint not easily recognisable.

2. Effect of isotropic mixing duration on ^{1}H photo-CIDNP selective TOCSY and Light-FESTA

Table S2 shows the experimental homonuclear (J_{HH}) and heteronuclear (J_{HF}) scalar coupling constants for 6-fluoroindole. The values in red are inter-ring J_{HH} coupling constants.

Table S2. Experimental scalar coupling constants (J / Hz) for 6FI. Values in red are the inter-ring J_{HH} couplings.

	H1	H2	H4	H5	F6	H7
H1	-	3.16	0	0	0	0.31
H2		-	0.37	0	0	0.87
H4			-	8.63	5.5	0.65
H5				-	9.5	2.43
F6					-	10.2
H7						-

Figure S3 shows the TOCSY build-up curves for transferring magnetisation from H2, which showed a 32-fold signal enhancement in the direct ^{1}H photo-CIDNP (Figure S2), to all the signals within the spin system in the ^{1}H photo-CIDNP selective TOCSY experiment. As $^3J_{H1H2}$ is 3.16 Hz, magnetisation transfers rapidly to H1 and oscillates between H1 and H2. The small inter-ring scalar coupling constants limit the effectiveness of TOCSY transfer to protons H4, H5 and H7, requiring a long isotropic mixing time (>200 ms) to achieve even small signal enhancements. Therefore, selective TOCSY did not provide satisfactory results for propagating the signal enhancement of H2 around the entire spin system.



Figure S3. Effect of isotropic mixing duration on 6FI ^{1}H signals in photo-CIDNP selective TOCSY experiments where H2 is selected. Each experiment had a 2 s illumination time during d1. Signal enhancements are corrected to account for loss of FMN due to photobleaching (see Section C5).

Figure S4 shows the TOCSY build-up curves for Light-FESTA experiments when (a) H4, (b) H7 and (c) H5 were selected. During the SRI-SME element, magnetisation is transferred from ^{19}F to the selected ^{1}H nuclei (as shown by 0 s isotropic mixing). Then, the magnetisation is transferred to the other ^{1}H nuclei within the spin system at a rate which is dependent on the J_{HH} couplings. For example, in Figure S4a, $^{3}J_{H4H5}$ has a value of 8.63 Hz so there is a relatively rapid transfer of magnetisation from H4 to H5, requiring only 50 ms isotropic mixing for almost complete transfer. As $^{5}J_{H4H7}$ has a much smaller value of 0.65 Hz, a longer mixing time (200 ms) is required to achieve optimal transfer of magnetisation to H7. The protons of the other ring (H1 and H2) show smaller enhancements as they have much smaller J_{HH} values. This suggests that a compromise mixing time may be used to achieve a more even enhancement of all signals within a spin system. Optimal transfer to the other ring protons occurs when ^{19}F magnetisation is transferred to H7 (Figure S4b). The optimal enhancement of H2 requires a 300 ms isotopic mixing time, while the optimal enhancement of H1 occurs at 250 ms. In each case, the total signal integral decreases as isotropic mixing time increases, due to relaxation during the DIPSI-2 element.



Figure S4. Effect of isotropic mixing duration on 6FI signals in Light-FESTA experiments. The SRI-SME element of FESTA selected (a) H4: 7.54 ppm, J_{HF} 5.5 Hz, (b) H7: 7.16 ppm, J_{HF} 10.2 Hz, and (c) H5: 6.86 ppm, J_{HF} 9.5 Hz. Integrals were corrected to account for FMN photobleaching during prolonged experiments (see Section C5).

As an alternative to using a compromise isotopic mixing duration, multiple experiments that have mixing times tailored to each signal can be acquired and the resultant spectra combined, to give good enhancements for all signals. Figure S5 shows the results from three Light-FESTA experiments that transferred ^{19}F magnetisation to (a) H4, (b) H5 and (c) H7; (d) shows the sum of these three Light-FESTA experiments (Σ-Light-FESTA).



Figure S5. Comparison of three Light-FESTA experiments obtained on a sample of 1 mM 6FI with 0.2 mM FMN. The three Light-FESTA experiments selected for (a) H4 J_{HF} 5.5 Hz, (b) H5 J_{HF} 9.5 Hz, and (c) H7 J_{HF} 10.2 Hz. Each experiment was acquired with 2 scans, illuminated for 2 s during the recovery delay and a 200 ms isotropic mixing duration. (d) Shows the sum of the three Light-FESTA spectra.

3. Experimental sensitivity enhancements for direct ^{1}H photo-CIDNP and Light-FESTA

Table S3 shows the signal-to-noise ratio (*i.e.*, experimental sensitivity) enhancements for direct ^{1}H photo-CIDNP, three Light-FESTA experiments transferring ^{19}F magnetisation to H4, H5 and H7, and for the sum of these three Light-FESTA experiments (Σ-Light-FESTA). The signal-to-noise ratio (SNR) for the Σ-Light-FESTA experiments has been corrected by a factor of $\sqrt{3}$ to account for the greater number of scans used. All signal-to-noise enhancements are relative to the same experiment conducted without illumination.

Table S3 clearly shows that there is a major sensitivity enhancement for multiple signals using the Light-FESTA approach, with >40-fold sensitivity enhancements for the H4-H5-H7 spin system using Σ -Light-FESTA. It should be noted that the SNR does not take into account the signal distortions present in direct ^{1}H photo-CIDNP spectra, as only the highest signal intensity in a multiplet is used, so the average sensitivity enhancements for H4, H5 and H7 multiplets are overestimated in this case.

Table S3: Sensitivity enhancements for direct ^{1}H photo-CIDNP, Light-FESTA, and Σ-Light-FESTA experiments using 6FI (1 mM) with FMN and 2 s of illumination. Sensitivity enhancements were calculated based on SNR and are relative to the same experiment conducted without illumination. Each of the Light-FESTA experiments selected one of the ^{1}H signals (H4, H5, or H7) coupled to ^{19}F. Σ-Light-FESTA provides the sum of these three experiments. All the values are corrected to account for different numbers of scans and are reported on a per-two-scan basis.

	H4	H1	H7	H5	H2
Direct ^{1}H photo-CIDNP	5.7	3.5	3.6	1.7	29.2
Light-FESTA H5 (6.86 ppm)	38.2	12.8	42.6	48.0	17.2
Light-FESTA H7 (7.16 ppm)	42.5	31.4	28.2	31.7	6.7
Light-FESTA H4 (7.54 ppm)	27.8	8.6	33.0	28.3	14.3
Σ-Light-FESTA	48.2	35.8	40.6	49.5	16.2

4. Evaluation of magnetisation transfer

The performance of various experiments for transferring magnetization from ^{19}F to ^{1}H was tested on a concentrated 6FI sample without illumination, to separate possible effects of photo-CIDNP from the effect of the pulse sequences themselves. Figure S6 shows the pulse sequences used for (a) Reverse-INEPT Modulated Echo (RI-ME),[6, 7] (b) RI-ME-TOCSY,[2, 5] (c) Selective Reverse-INEPT Selective Modulated Echo (SRI-SME)[2, 5] and (d) SRI-SME-TOCSY (*i.e.*, FESTA)[2] experiments.



Figure S6. Schematic representations of (a) Reverse-INEPT Modulated Echo (RI-ME), (b) RI-ME-TOCSY, (c) Selective Reverse-INEPT Selective Modulated Echo (SRI-SME) and (d) SRI-SME-TOCSY (*i.e.*, FESTA experiment) pulse sequences. Narrow black and grey rectangles represent hard 90° and 180° pulses, respectively. The white shaped pulses represent selective 180° pulses. Trapezoids with diagonal arrows represent chirp pulses used to suppress zero-quantum coherences. Light grey trapezoids represent pulsed field gradients: G_1, G_2, G_4 and G_5 are used for coherence transfer pathway selection; G_3 and G_7 are homospoil gradients; and G_6 and G_8 are weak rectangular gradients applied simultaneously with the chirp pulses.

Figure S7 shows the results obtained using each of these experiments on a 100 mM sample of 6FI in D_2O. Using hard 180° ^{1}H pulses, as in the RI-ME experiment (Figure S6a), leads to phase-modulated multiplets in the resulting spectrum (Figure S7a) due to J_{HH} modulation during Δ delays. Incorporating a TOCSY element, as in the RI-ME-TOCSY experiment (Figure S6b), purges the phase modulation caused by J_{HH} but incurs a large sensitivity penalty (Figure S7b). Compared to the RI-ME data, utilising selective 180° ^{1}H pulses in the SRI-SME experiment (Figure S6c) avoids J_{HH} modulation while providing a significant sensitivity improvement (Figure S7c). Finally, the SRI-SME-TOCSY (FESTA) experiment (Figure S6d) resulted in improved sensitivity for all ^{1}H signals within the same spin system as the chosen ^{19}F nucleus (Figure S7d) compared to the RI-ME-TOCSY data. Thus, the SRI-SME-TOCSY pulse sequence was chosen as the basis for Light-FESTA.



Figure S7: Comparison of the results obtained on a 100 mM sample of 6FI in D_2O using hard 180° pulses in (a) RI-ME and (b) RI-ME-TOCSY compared to using selective 180° ^{1}H pulses in (c) SRI-SME and (d) SRI-SME-TOCSY (FESTA) experiments selecting H7 (7.18 ppm) of 6FI. The RI-ME and RI-ME-TOCSY data were acquired with a compromise Δ value corresponding to J_{HF} = 8.5 Hz. The SRI-SME and FESTA experiments were acquired with a Δ value corresponding to J_{HF} = 10.2 Hz. All experiments were acquired with 4 transients, 4 dummy scans and a recovery delay of 5 s. The RI-ME-TOCSY and FESTA experiments used a DIPSI-2 isotropic mixing element of 150 ms. Data were acquired on a Bruker 500 MHz NEO spectrometer using a 5 mm TBI probe equipped with a z-gradient coil with a maximum nominal gradient strength of 0.67 T m^{-1}. Here, no illumination was applied during the recovery delay d1. All spectra are scaled consistently.

5. Effect of illumination time on Light-FESTA signals

Figure S8 shows the illumination build-up curves for the Light-FESTA experiment when (a) H4, (b) H7 and (c) H5 are selected in the SRI-SME element and a 200 ms isotropic mixing duration is used. In all cases at 0 s illumination, there is no signal enhancement. As the illumination time increases, the signal enhancement in Light-FESTA also increases, up to a maximum at 4 s. After this there is no further signal enhancement on increasing the illumination time, as the rate of polarisation is balanced out by longitudinal relaxation.[8] No further signal intensity variations are expected except those from photodegradation (e.g. photobleaching) effects. Figure S8a shows a smaller total signal enhancement than Figures S8b and S8c, which are similar. As J_{H4F} (5.5 Hz) is smaller than both J_{H5F} and J_{H7F} (9.5 Hz and 10.2 Hz, respectively) the Δ delay in Figure S1 is longer for magnetisation transfer to H4 than it is for H5 and H7. Hence, the reduction in signal enhancement may be attributed to transverse relaxation during this delay. This is further evidenced by Figure S4, which shows that in the absence of isotropic mixing the signal enhancement of H4 is significantly less than that of H5 or H7.



Figure S8. Effect of illumination time on Light-FESTA experiments using 200 ms isotropic mixing duration. The SRI-SME element of FESTA selected (a) H4: 7.54 ppm, J_{HF} 5.5 Hz, (b) H7: 7.16 ppm, J_{HF} 10.2 Hz, and (c) H5: 6.86 ppm, J_{HF} 9.5 Hz. Integrals are corrected to account for FMN photobleaching during the prolonged experiments.

During the investigation, it was noticed that with prolonged illumination of the same sample photo-CIDNP signal enhancements were gradually reduced, due to the photobleaching of FMN. Measuring ^{19}F photo-CIDNP signal enhancements as a function of total illumination time showed a smooth decay with time (Figure S9). Therefore, by monitoring the total exposure of a sample to light, the effect of photobleaching on signal intensity can be corrected to make the effects of other experimental parameters clearer. Bleaching-corrected values of signal intensity are reported throughout this work.



Figure S9. Effect of total illumination time on ^{19}F signal enhancements. 32 direct ^{19}F spectra were acquired consecutively, each using 2 s illumination during the recovery delay. All experiments were acquired with 1 scan, 32k points, a 50 ppm spectral width and a recovery delay of 13 s.

D. References

1. J. E. Bramham and A. P. Golovanov, *Commun. Chem.*, 2022, **5**, 90.
2. L. Castañar, P. Moutzouri, T. M. Barbosa, C. F. Tormena, R. Rittner, A. R. Phillips, S. R. Coombes, M. Nilsson and G. A. Morris, *Anal. Chem.*, 2018, **90**, 5445-5450.
3. J. Cavanagh and M. Rance, *J. Magn. Reson. (1969)*, 1992, **96**, 670-678.
4. M. J. Thrippleton and J. Keeler, *Angew. Chem. Int. Ed.*, 2003, **42**, 3938-3941.

5. A. J. Shaka and R. Freeman, J. Magn. Reson.,1982,502-507.
6. T. M. Barbosa, L. Castañar, P. Moutzouri, M. Nilsson, G. A. Morris, R. Rittner and C. F. Tormena, *Anal. Chem.*, 2020, **92**, 2224-2228.
7. G. A. Morris and R. Freeman, J. Am. Chem. Soc., 1979.
8. A. Sekhar and S. Cavagnero, *J. Phys. Chem. B*, 2009, **113**, 8310-8318.

E. Pulse sequences in Bruker format

All pulse sequence codes can be downloaded directly from the data repository (DOI: 10.48420/23056013). Alternatively, the pulse sequence codes may be downloaded from https://nmr.manchester.ac.uk

1. Light-FESTA

```
; Doubly selective refocused INEPT - TOCSY (19F - 1H correlation and 1H-1H TOCSY transfer)
;    using selective 90 19F pulse for excitation
;    using  selective 180 19F pulse for refocusing (optional)
;    using  selective 180 1H pulse for refocusing
;    using PFG for CTP selection
;    homonuclear Hartman-Hahn transfer using DIPSI2 sequence for mixing with zero quantum
suppression (optional)
; Illumination of the sample is applied during the recovery delay

; Developed by:
; NMR Methodology Group
; University of Manchester
; Marshall Smith
; Jack Bramham
;
; Options:
; [cnst1 = 0] without 1H purge pulses before d1
; [cnst1 = 1] with 1H purge pulses before d1

; REFERENCES
; L. Castanar, P. Moutzouri, T. Barbosa, C. Tormena, R. Rittner, A. Phillips, S. Coombes, M. Nilsson, G. A.
;Morris., Anal. Chem., 2018, 90, 8, 5445-5450
; M.J. Thrippleton & J. Keeler, Angew. Chem. Int. Ed. 42, 3938-3941 (2003)
; J. E. Bramham and A. P. Golovanov, Commun. Chem., 2022, 5, 90.

;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define loopcounter PWMCounter
define delay OnDelay
define delay OffDelay

"l7=cnst61"
"OnDelay = (cnst61/100)*d31"
"OffDelay = d31-OnDelay"
"PWMCounter = d30 / d31"

"d11=30m+1s/(cnst12)-1s/(cnst12)"
"d11=30m+1s/(cnst51)-1s/(cnst51)"
"d11=30m"
"d2=1s/(cnst2*4*cnst3)"
"d3=1s/(cnst2*4*cnst7)"
```

```
"d13=20u"
"d16=200u"

"DELTA1=d2-p16-d16-larger(p12,p4)/2"
"DELTA2=d2-2*p16-2*d16-larger(p12,p4)/2"
"DELTA3=d3-2*d16-2*p16-larger(p12,p4)/2"
"DELTA4=d3-d16-p16-larger(p12,p4)/2"
"p2=p1*2"
"p4=p3*2"
"p16=1m"
"spoff11=0"
;"spoff12=bf1*(cnst12/1000000)-o1"
"spoff15=0"
"spoff29=0"

;; WAVEMAKER;;
;sp12(p12):wvm:rsnob_wvm:f1 rsnob(cnst51 Hz, cnst12 ppm; NPOINTS=1000; PHI=0.5)

"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"
"acqt0=-p1*2/3.1416"

1 ze
 30m pl12:f2
2 30m do:f2
 d13 BLKGRAMP
 if "cnst1 == 1" ;Purge pulses
 {
 d13 UNBLKGRAMP
 d13 pl1:f1
 ;p16:gp20
 ;d18
 (p1 ph20):f1
 (p1 ph21):f1
 p16:gp20

 d18
 d13 BLKGRAMP
 }
 else
 {
 }

  50u BLKGRAMP
  50u LOCKH_OFF
  d1 pl1:f1 pl2:f2
  50u LOCKH_ON
  50u UNBLKGRAMP

if (l7==0) ; no light loop
{
  0.05u
  d30 ;delay with no illumination
  0.05u
```

```
  goto 4 ;skip light loops and go to pulse and acquisition
}
  if (l7==100) ; full brightness loop
{
  0.05u setnmr3|10;light turned on
  d30; light on during this delay
  0.05u setnmr3^10;light turned off
}
else ; pulse width modulation loop
{
  "OnDelay = (cnst61/100)*d31"
  "OffDelay = d31-OnDelay"
  "PWMCounter = d30 / d31"
3 0.05u setnmr3|10;light turned on
  OnDelay ; light on during this delay
  0.05u setnmr3^10;light turned off
  OffDelay ; light is off during this delay
  lo to 3 times PWMCounter
}

4 (p3 ph1):f2 ;F - H AP transfer

 DELTA1 pl0:f1 pl2:f2 ;pl2:f2
 p16:gp5
 d16
(center (p12:sp12 ph4):f1 (p4 ph3):f2 )
 p16:gp5
 d16
 DELTA2 pl1:f1 pl2:f2
 p16:gp1
 d16

 (p3 ph4):f2
 p16:gp2
 d16
 (p1 ph5):f1

p16:gp1*0.9407
 d16

 DELTA3 pl0:f1
 p16:gp3
 d16
 (center (p12:sp12 ph6):f1 (p4 ph7):f2 )
 p16:gp3
 d16 pl1:f1
 DELTA4

(center (p1 ph8):f1 (p3 ph11):f2 )

 10u gron0 pl0:f1
 (p32:sp29 ph9):f1
 10u groff
 d16 pl10:f1
                                ;begin DIPSI2
```

```
5 p6*3.556 ph23
 p6*4.556 ph25
 p6*3.222 ph23
 p6*3.167 ph25
 p6*0.333 ph23
 p6*2.722 ph25
 p6*4.167 ph23
 p6*2.944 ph25
 p6*4.111 ph23
 p6*3.556 ph25
 p6*4.556 ph23
 p6*3.222 ph25
 p6*3.167 ph23
 p6*0.333 ph25
 p6*2.722 ph23
 p6*4.167 ph25
 p6*2.944 ph23
 p6*4.111 ph25
 p6*3.556 ph25
 p6*4.556 ph23
 p6*3.222 ph25
 p6*3.167 ph23
 p6*0.333 ph25
 p6*2.722 ph23
 p6*4.167 ph25
 p6*2.944 ph23
 p6*4.111 ph25
 p6*3.556 ph23
 p6*4.556 ph25
 p6*3.222 ph23
 p6*3.167 ph25
 p6*0.333 ph23
 p6*2.722 ph25
 p6*4.167 ph23
 p6*2.944 ph25
 p6*4.111 ph23
 lo to 5 times l1
;end DIPSI2
 p16:gp4
 d16 pl0:f1
 10u gron0*1.333
 (p32*0.75:sp29 ph9):f1
 10u groff
 d16 pl1:f1 pl2:f2 BLKGRAMP
 (p1 ph10):f1

 20u pl12:f2
 go=2 ph31 cpd2:f2
 30m do:f2 mc #0 to 2 F0(zd)
exit
            50u LOCKH_OFF

ph1=0 2 2 0
ph2=0
ph3=0
```

```
ph4=1
ph5=0
ph6=0
ph7=0
ph8=1
ph9=0
ph10=0
ph11=0
ph20=0
ph21=1
ph23=3
ph25=1
ph31=0 2 2 0

;POWER LEVEL
;pl0 : zero power (0 W)
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl10 : f1 channel - power level for TOCSY-spinlock
;pl12 : f2 channel - power level for CPD/BB decoupling
;sp11 : f2 channel - power level of excitation shaped pulse
;sp12 : f1 channel - power level of refocusing shaped pulse
;sp12 : f2 channel - power level of refocusing shaped pulse
;sp29 : f1 channel - power level of the adiabatic pulse of ZQF element

; PULSE DURATION
;p1 : f1 channel – 90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel - 90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p6 : f1 channel - 90 degree low power pulse for DIPSI element
;p11: f2 channel - 90 degree low power shaped pulse
; choose p11 according to the desired selectivity
;p12: f1 channel - 180 degree low power shaped pulse
; choose p12 according to the desired selectivity
;p13: f2 channel - 180 degree low power shaped pulse
; choose p13 according to the desired selectivity
;p32: f1 channel - 180 degree shaped pulse (adiabatic) [20 msec]

; GRADIENT DURATION
;p16: duration of CTP gradients [1 msec]

; DELAY
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(nH*4*J(XH))
;d3 : 1/(nF*4*J(XH))
;d9 : TOCSY mixing time
;d16: delay for gradient recovery [200 usec]
;d30 : illumination time
;d31 : cycle time

; PULSE SHAPE
;spnam11 : f2 channel - file name for the selective 90 excitation shaped pulse [Gauss]
;spnam12 : f1 channel - file name for the selective 180 refocusing shaped pulse [Rsnob]
;spnam13 : f2 channel - file name for the selective 180 refocusing shaped pulse [Rsnob]
```

;spnam29 : f1 channel - file name for the adiabatic shaped pulse using in ZQF [Crp60,20,20.10]
; smoothed chirp (sweepwidth, 20% smoothing, 10000 points)

;GRADIENT SHAPE
;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam3 : SMSQ10.100
;gpnam4 : SMSQ10.100
;gpnam5 : SMSQ10.100
;gpnam20: SMSQ10.100
;gpnam21: SMSQ10.100

;GRADIENT STRENGTH
;gpz0 : ZQF gradient [3%]
;gpz1 : F-H CTP gradient [47%]
;gpz2 : homospoil gradient [37%]
;gpz3 : 180 1H CTP gradient [41%]
;gpz4 : homospoil gradient TOCSY [17%]
;gpz5 : 180 1H CTP gradient [33%]
;gpz20 : homospoil gradient purge pulses [53%]
;gpz21 : homospoil gradient purge pulses [59%]

;CONSTANTS
;cnst1 : 0 (--) / 1 (Purge pulses)
;cnst2 : J(XH)
;cnst3 : number of protons coupled with fluorine
;cnst7 : number of fluorines coupled with protons
;cnst6 : 0 (--) / 1 (ZQF-TOCSY)
;cnst12: chemical shift for selective 1H pulse (offset, in ppm)
;cnst51 : Bandwidth of 1H selective pulse
;cnst61 : light duty-cycle (% brightness)

;OTHER
;ns: 2 * n, total number of scans: NS * TD0
;ds: 2
;set O2 on resonance to the 19F signal of interest
;cpd2: p5m4sp180
;l1: loop for DIPSI cycle: ((p6*115.112) * l1) = mixing time
;if any DELTA become negative try to reduce the gradient duration or selective pulse duration

2. photo-CIDNP selective TOCSY with variable isotropic mixing duration

```
;1D homonuclear Hartman-Hahn transfer using
;   DIPSI2 sequence for mixing
;   using selective refocusing with a shaped pulse
;with zero quantum suppression
; Selective pulse can be generated using wavemaker

; Developed by:
; NMR Methodology Group
; University of Manchester
; Marshall Smith, Jack Bramham
;
; Options:
; [cnst1 = 0] without 1H purge pulses before d1
; [cnst1 = 1] with 1H purge pulses before d1

; REFERENCES
;H. Kessler, H. Oschkinat, C. Griesinger & W. Bermel, J. Magn. Reson. 70, 106 (1986)
;J. Stonehouse, P. Adell, J. Keeler & A.J. Shaka, J. Am. Chem. Soc 116, 6037 (1994)
;K. Stott, J. Stonehouse, J. Keeler, T.L. Hwang & A.J. Shaka, J. Am. Chem. Soc 117, 4199-4200 (1995)
;M.J. Thrippleton & J. Keeler, Angew. Chem. Int. Ed. 42, 3938-3941 (2003)

;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define loopcounter PWMCounter
define delay OnDelay
define delay OffDelay

"l7=cnst61"
"OnDelay = (cnst61/100)*d31"
"OffDelay = d31-OnDelay"
"PWMCounter = d30 / d31"

"d11=30m+1s/(cnst12)-1s/(cnst12)"
"d11=30m+1s/(cnst51)-1s/(cnst51)"
"d11=30m"
"spoff29=0"
"spoff30=0"

;Wavemaker
;sp12(p12):wvm:rsnob_wvm:f1 rsnob(cnst51 Hz, cnst12 ppm; NPOINTS=1000; PHI=0.5)

"acqt0=-p1*2/PI"

1 ze
2 30m
```

```
"d9=vd"
"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"

  20u pl1:f1 BLKGRAD
  d1

if (l7==0) ; no light loop
{
  0.05u
  d30 ;delay with no illumination
  0.05u
  goto 4 ;skip light loops and go to pulse and acquisition
}
  if (l7==100) ; full brightness loop
{
  0.05u setnmr3|10;light turned on
  d30; light on during this delay
  0.05u setnmr3^10;light turned off
}
else ; pulse width modulation loop
{
  "OnDelay = (cnst61/100)*d31"
  "OffDelay = d31-OnDelay"
  "PWMCounter = d30 / d31"
3 0.05u setnmr3|10; light turned on
  OnDelay ; light on during this delay
  0.05u setnmr3^10; light turned off
  OffDelay ; light is off during this delay
  lo to 3 times PWMCounter
}

  50u UNBLKGRAD
  (p1 ph1):f1
  3u
  p16:gp1
  d16 pl0:f1
  p12:sp12:f1 ph2:r
  3u
  p16:gp1
  d16 pl1:f1

  p1 ph3
  10u gron0
  (p32:sp29 ph3):f1
  20u groff
  d16 pl10:f1
 p16:gp3
                                                    ;begin DIPSI2
4 p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
```

```
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23

  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25

  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25

  p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23
  lo to 4 times l1
                                                    ;end DIPSI2
  p16:gp2
  d16
  10u gron0*1.333
  (p32*0.75:sp29 ph3):f1
  20u groff
  d16 pl1:f1
  p1 ph3
  go=2 ph31
  30m mc #0 to 2 F1QF(ivd)
  20u BLKGRAD
exit

ph1=0 2
ph2=0 0 1 1 2 2 3 3
ph3=0
ph23=3
ph25=1
ph31=0 2 2 0
```

```
        ; POWER LEVEL
        ;pl0 : 0W
        ;pl1 : f1 channel - power level for pulse (default)
        ;pl10: f1 channel - power level for TOCSY-spinlock
        ;sp2: f1 channel - shaped pulse
        ;sp29: f1 channel - shaped pulse (adiabatic)

        ; PULSE DURATION
        ;p1 : f1 channel -  90 degree high power pulse
        ;p6 : f1 channel -  90 degree low power pulse
        ;p12: f1 channel - 180 degree shaped pulse
        ;p16: homospoil/gradient pulse                    [1 msec]
        ;p32: f1 channel - 180 degree shaped pulse (adiabatic)    [20 msec]
        ;     smoothed chirp (sweepwidth, 20% smoothing, 10000 points)
        ;p34: f1 channel - 180 degree shaped pulse (adiabatic)    [15 msec]
        ;     smoothed chirp (sweepwidth, 20% smoothing, 10000 points)

        ; DELAYS
        ;d1 : relaxation delay; 1-5 * T1
        ;d9 : TOCSY mixing time
        ;d16: delay for homospoil/gradient recovery
        ;d30 : illumination time
        ;d31 : cycle time

        ;GRADIENT STRENGTH
        ;gpz0: ca. 11%
        ;gpz1: 15%
        ;gpz2: 31%

        ;GRADIENT SHAPE
        ;gpnam1: SMSQ10.100
        ;gpnam2: SMSQ10.100

        ; CONSTANTS
        ;cnst12 : Chemical shift for 1H selective pulse
        ;cnst51 : Bandwidth of 1H selective pulse
        ;cnst61: light duty-cycle (% brightness)

        ; Other
        ;l1: loop for DIPSI cycle: ((p6*115.112) * l1) = mixing time
        ;ns: 8 * n, total number of scans: NS * TD0
        ;ds: 4
        ;phcor 2 : phase difference between power levels sp1 and pl1
        ;for sweepwidth of adiabatic shape and adjusting gpz0
;see supplementary material of M.J. Thrippleton & J. Keeler, Angew. Chem. Int. Ed. 42, 3938-3941 (2003)
```

3. Light-FESTA with variable illumination time

```
; Doubly selective refocused INEPT - TOCSY (19F - 1H correlation and 1H-1H TOCSY transfer)
;   using selective 90 19F pulse for excitation
;   using  selective 180 19F pulse for refocusing (optional)
;   using  selective 180 1H pulse for refocusing
;   using PFG for CTP selection
;   homonuclear Hartman-Hahn transfer using DIPSI2 sequence for mixing with zero quantum
  suppression (optional)
; Variable illumination times should be defined in the variable delay list (vdlist)

; Developed by:
; NMR Methodology Group
; University of Manchester
; Marshall Smith
; Jack Bramham
;
; Options:
; [cnst1 = 0] without 1H purge pulses before d1
; [cnst1 = 1] with 1H purge pulses before d1

; REFERENCES
; L. Castañar, P. Moutzouri, T. Barbosa, C. Tormena, R. Rittner, A. Phillips, S. Coombes, M. Nilsson, G. A.
;Morris., Anal. Chem., 2018, 90, 8, 5445-5450
; M.J. Thrippleton & J. Keeler, Angew. Chem. Int. Ed. 42, 3938-3941 (2003)
; J. E. Bramham and A. P. Golovanov, Commun. Chem., 2022, 5, 90.

;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define loopcounter PWMCounter
define delay OnDelay
define delay OffDelay

"l7=cnst61"
"OnDelay = (cnst61/100)*d31"
"OffDelay = d31-OnDelay"
"PWMCounter = d30 / d31"

"d11=30m+1s/(cnst12)-1s/(cnst12)"
"d11=30m+1s/(cnst51)-1s/(cnst51)"

"d11=30m"
"d30=0"
```

```
"d2=1s/(cnst2*4*cnst3)"
"d3=1s/(cnst2*4*cnst7)"

"d13=20u"
"d16=200u"

"DELTA1=d2-p16-d16-larger(p12,p4)/2"
"DELTA2=d2-2*p16-2*d16-larger(p12,p4)/2"
"DELTA3=d3-2*d16-2*p16-larger(p12,p4)/2"
"DELTA4=d3-d16-p16-larger(p12,p4)/2"
"p2=p1*2"
"p4=p3*2"
"p16=1m"

;OFFSETS
"spoff11=0"
"spoff15=0"
"spoff29=0"

;; WAVEMAKER;;
;cnst51 : Bandwidth of 1H selective pulse
;cnst12 : Chemical shift for 1H selective pulse
;sp12(p12):wvm:rsnob_wvm:f1 rsnob(cnst51 Hz, cnst12 ppm; NPOINTS=1000; PHI=0.5)

"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"

"acqt0=-p1*2/3.1416"

1 ze
 30m pl12:f2
2 30m do:f2
 d13 BLKGRAMP


"d30=vd"
"l7=cnst61"
"OnDelay = (cnst61/100)*d31"
"OffDelay = d31-OnDelay"
"PWMCounter = d30 / d31"

 if "cnst1 == 1" ;Purge pulses
 {
 d13 UNBLKGRAMP
 d13 pl1:f1
 ;p16:gp20
 ;d18
 (p1 ph20):f1
 (p1 ph21):f1
 p16:gp20
```

```
 d18
 d13 BLKGRAMP
 }
 else
 {
 }

  50u BLKGRAMP
  50u LOCKH_OFF
  d1 pl1:f1 pl2:f2
  50u LOCKH_ON
  50u UNBLKGRAMP

if (l7==0) ; no light loop
{
  0.05u
  d30 ;delay with no illumination
  0.05u
  goto 4 ;skip light loops and go to pulse and acquisition
}
  if (l7==100) ; full brightness loop
{
  if "d30==0" goto 4
  0.05u setnmr3|10;light turned on
  d30; light on during this delay
  0.05u setnmr3^10;light turned off
}
else ; pulse width modulation loop
{
  "OnDelay = (cnst61/100)*d31"
  "OffDelay = d31-OnDelay"
  "PWMCounter = d30 / d31"
3 0.05u setnmr3|10;light turned on
  OnDelay ; light on during this delay
  0.05u setnmr3^10;light turned off
  OffDelay ; light is off during this delay
  lo to 3 times PWMCounter
}

4 (p3 ph1):f2 ;F - H AP transfer

 DELTA1 pl0:f1 pl2:f2 ;pl2:f2
 p16:gp5
 d16
(center (p12:sp12 ph4):f1 (p4 ph3):f2 )
 p16:gp5
 d16
 DELTA2 pl1:f1 pl2:f2
 p16:gp1
 d16
```

(p3 ph4):f2
p16:gp2
d16
(p1 ph5):f1

p16:gp1*0.9407
d16

DELTA3 pl0:f1
p16:gp3
d16
(center (p12:sp12 ph6):f1 (p4 ph7):f2)
p16:gp3
d16 pl1:f1
DELTA4

(center (p1 ph8):f1 (p3 ph11):f2)

10u gron0 pl0:f1
(p32:sp29 ph9):f1
10u groff
d16 pl10:f1

5 p6*3.556 ph23 ;begin DIPSI2
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25
p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25
p6*3.556 ph23

```
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
lo to 5 times l1 ;end DIPSI2

p16:gp4
d16 pl0:f1
10u gron0*1.333
(p32*0.75:sp29 ph9):f1
10u groff
d16 pl1:f1 pl2:f2 BLKGRAMP
(p1 ph10):f1



20u pl12:f2
go=2 ph31 cpd2:f2
30m do:f2 mc #0 to 2 F1QF(ivd)

exit
        50u LOCKH_OFF

ph1=0 2 2 0
ph2=0
ph3=0
ph4=1
ph5=0
ph6=0
ph7=0
ph8=1
ph9=0
ph10=0
ph11=0
ph20=0
ph21=1
ph23=3
ph25=1
ph31=0 2 2 0

;POWER LEVEL
;pl0 : zero power (0 W)
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl10 : f1 channel - power level for TOCSY-spinlock
;pl12 : f2 channel - power level for CPD/BB decoupling
;sp11 : f2 channel - power level of excitation shaped pulse
```

;sp12 : f1 channel - power level of refocusing shaped pulse
;sp12 : f2 channel - power level of refocusing shaped pulse
;sp29 : f1 channel - power level of the adiabatic pulse of ZQF element

;PULSE DURATION
;p1 : f1 channel - 90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel - 90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p6 : f1 channel - 90 degree low power pulse for DIPSI element
;p11: f2 channel - 90 degree low power shaped pulse
; choose p11 according to the desired selectivity
;p12: f1 channel - 180 degree low power shaped pulse
; choose p12 according to the desired selectivity
;p13: f2 channel - 180 degree low power shaped pulse
; choose p13 according to the desired selectivity
;p32: f1 channel - 180 degree shaped pulse (adiabatic) [20 msec]

;GRADIENT DURATION
;p16: duration of CTP gradients [1 msec]

;DELAY
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(nH*4*J(XH))
;d3 : 1/(nF*4*J(XH))
;d9 : TOCSY mixing time
;d16: delay for gradient recovery [200 usec]
;d30 : illumination time
;d31 : cycle time

;PULSE SHAPE
;spnam12 : f1 channel - file name for the selective 180 refocusing shaped pulse [Rsnob]
;spnam13 : f2 channel - file name for the selective 180 refocusing shaped pulse [Rsnob]
;spnam29 : f1 channel - file name for the adiabatic shaped pulse using in ZQF [Crp60,20,20.10]
; smoothed chirp (sweepwidth, 20% smoothing, 10000 points)

;GRADIENT SHAPE
;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam3 : SMSQ10.100
;gpnam4 : SMSQ10.100
;gpnam5 : SMSQ10.100
;gpnam20: SMSQ10.100
;gpnam21: SMSQ10.100

;GRADIENT STRENGTH
;gpz0 : ZQF gradient [3%]
;gpz1 : F-H CTP gradient [47%]
;gpz2 : homospoil gradient [37%]
;gpz3 : 180 1H CTP gradient [41%]
;gpz4 : homospoil gradient TOCSY [17%]

;gpz5 : 180 1H CTP gradient [33%]
;gpz20 : homospoil gradient purge pulses [53%]
;gpz21 : homospoil gradient purge pulses [59%]

;CONSTANTS
;cnst1 : 0 (--) / 1 (Purge pulses)
;cnst2 : J(XH)
;cnst3 : number of protons coupled with fluorine
;cnst7 : number of fluorines coupled with protons
;cnst6 : 0 (--) / 1 (ZQF-TOCSY)
;cnst12: chemical shift for selective 1H pulse (offset, in ppm)
;cnst51 : Bandwidth of 1H selective pulse
;cnst61 : light duty-cycle (% brightness)

;OTHER
;ns: 2 * n, the total number of scans: NS * TD0
;ds: 2
;set O2 on resonance to the 19F signal of interest
;cpd2: p5m4sp180
;l1: loop for DIPSI cycle: ((p6*115.112) * l1) = mixing time
;if any DELTA become negative try to reduce the gradient duration or selective pulse duration

4. Light-FESTA with variable TOCSY mixing time

```
; Doubly selective refocused INEPT - TOCSY (19F - 1H correlation and 1H-1H TOCSY transfer)
;   using selective 90 19F pulse for excitation
;   using  selective 180 19F pulse for refocusing (optional)
;   using  selective 180 1H pulse for refocusing
;   using PFG for CTP selection
;   homonuclear Hartman-Hahn transfer using DIPSI2 sequence for mixing with zero quantum
  suppression (optional)
; Variable isotropic mixing times should be defined in the variable delay list (vdlist)


; Developed by:
; NMR Methodology Group
; University of Manchester
; Marshall Smith
; Jack Bramham
;
; Options:
; [cnst1 = 0] without 1H purge pulses before d1
; [cnst1 = 1] with 1H purge pulses before d1


; REFERENCES
; L. Castañar, P. Moutzouri, T. Barbosa, C. Tormena, R. Rittner, A. Phillips, S. Coombes, M. Nilsson, G. A.
;Morris., Anal. Chem., 2018, 90, 8, 5445-5450
; M.J. Thrippleton & J. Keeler, Angew. Chem. Int. Ed. 42, 3938-3941 (2003)
; J. E. Bramham and A. P. Golovanov, Commun. Chem., 2022, 5, 90.

;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define loopcounter PWMCounter
define delay OnDelay
define delay OffDelay

"l7=cnst61"
"OnDelay = (cnst61/100)*d31"
"OffDelay = d31-OnDelay"
"PWMCounter = d30 / d31"

"d11=30m+1s/(cnst12)-1s/(cnst12)"
"d11=30m+1s/(cnst51)-1s/(cnst51)"

"d11=30m"
```

```
"d2=1s/(cnst2*4*cnst3)"
"d3=1s/(cnst2*4*cnst7)"

"d13=20u"
"d16=200u"

"DELTA1=d2-p16-d16-larger(p12,p4)/2"
"DELTA2=d2-2*p16-2*d16-larger(p12,p4)/2"
"DELTA3=d3-2*d16-2*p16-larger(p12,p4)/2"
"DELTA4=d3-d16-p16-larger(p12,p4)/2"
"p2=p1*2"
"p4=p3*2"
"p16=1m"
"spoff11=0"
;"spoff12=bf1*(cnst12/1000000)-o1"
"spoff15=0"
"spoff29=0"

;; WAVEMAKER;;
;sp12(p12):wvm:rsnob_wvm:f1 rsnob(cnst51 Hz, cnst12 ppm; NPOINTS=1000; PHI=0.5)

"acqt0=-p1*2/3.1416"

1 ze
 30m pl12:f2
2 30m do:f2
 d13 BLKGRAMP

"d9=vd"
"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"

 if "cnst1 == 1" ;Purge pulses
 {
 d13 UNBLKGRAMP
 d13 pl1:f1
 ;p16:gp20
 ;d18
 (p1 ph20):f1
 (p1 ph21):f1
 p16:gp20

 d18
 d13 BLKGRAMP
 }
 else
 {
 }


  50u BLKGRAMP
```

```
  50u LOCKH_OFF
  d1 pl1:f1 pl2:f2
  50u LOCKH_ON
  50u UNBLKGRAMP

if (l7==0) ; no light loop
{
  0.05u
  d30 ;delay with no illumination
  0.05u
  goto 4 ;skip light loops and go to pulse and acquisition
}
  if (l7==100) ; full brightness loop
{
  0.05u setnmr3|10;light turned on
  d30; light on during this delay
  0.05u setnmr3^10;light turned off
}
else ; pulse width modulation loop
{
  "OnDelay = (cnst61/100)*d31"
  "OffDelay = d31-OnDelay"
  "PWMCounter = d30 / d31"
3 0.05u setnmr3|10;light turned on
  OnDelay ; light on during this delay
  0.05u setnmr3^10;light turned off
  OffDelay ; light is off during this delay
  lo to 3 times PWMCounter
}

4 (p3 ph1):f2 ;F - H AP transfer

 DELTA1 pl0:f1 pl2:f2 ;pl2:f2
 p16:gp5
 d16
(center (p12:sp12 ph4):f1 (p4 ph3):f2 )
 p16:gp5
 d16
 DELTA2 pl1:f1 pl2:f2
 p16:gp1
 d16

 (p3 ph4):f2
 p16:gp2
 d16
 (p1 ph5):f1

p16:gp1*0.9407
 d16

 DELTA3 pl0:f1
```

```
 p16:gp3
 d16
 (center (p12:sp12 ph6):f1 (p4 ph7):f2 )
 p16:gp3
 d16 pl1:f1
 DELTA4

(center (p1 ph8):f1 (p3 ph11):f2 )

 10u gron0 pl0:f1
 (p32:sp29 ph9):f1
 10u groff
 d16 pl10:f1

p16:gp6

5 p6*3.556 ph23          ;begin DIPSI2
 p6*4.556 ph25
 p6*3.222 ph23
 p6*3.167 ph25
 p6*0.333 ph23
 p6*2.722 ph25
 p6*4.167 ph23
 p6*2.944 ph25
 p6*4.111 ph23
 p6*3.556 ph25
 p6*4.556 ph23
 p6*3.222 ph25
 p6*3.167 ph23
 p6*0.333 ph25
 p6*2.722 ph23
 p6*4.167 ph25
 p6*2.944 ph23
 p6*4.111 ph25
 p6*3.556 ph25
 p6*4.556 ph23
 p6*3.222 ph25
 p6*3.167 ph23
 p6*0.333 ph25
 p6*2.722 ph23
 p6*4.167 ph25
 p6*2.944 ph23
 p6*4.111 ph25
 p6*3.556 ph23
 p6*4.556 ph25
 p6*3.222 ph23
 p6*3.167 ph25
 p6*0.333 ph23
 p6*2.722 ph25
 p6*4.167 ph23
 p6*2.944 ph25
```

```
 p6*4.111 ph23
 lo to 5 times l1              ;end DIPSI2

 p16:gp4
 d16 pl0:f1
 10u gron0*1.333
 (p32*0.75:sp29 ph9):f1
 10u groff
 d16 pl1:f1 pl2:f2
 (center (p1 ph10):f1 (p3 ph11):f2)

 50u pl12:f2 BLKGRAMP
 go=2 ph31 cpd2:f2
 30m do:f2 mc #0 to 2 F1QF(ivd)
exit
          50u LOCKH_OFF

ph1=0 2 2 0
ph2=0
ph3=0
ph4=1
ph5=0
ph6=0
ph7=0
ph8=1
ph9=0
ph10=0
ph11=0
ph20=0
ph21=1
ph23=3
ph25=1
ph31=0 2 2 0

;POWER LEVEL
;pl0 : zero power (0 W)
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl10 : f1 channel - power level for TOCSY-spinlock
;pl12 : f2 channel - power level for CPD/BB decoupling
;sp11 : f2 channel - power level of excitation shaped pulse
;sp12 : f1 channel - power level of refocusing shaped pulse
;sp12 : f2 channel - power level of refocusing shaped pulse
;sp29 : f1 channel - power level of the adiabatic pulse of ZQF element

;PULSE DURATION
;p1 : f1 channel - 90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel - 90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p6 : f1 channel - 90 degree low power pulse for DIPSI element
```

```
;p11: f2 channel - 90 degree low power shaped pulse
; choose p11 according to the desired selectivity
;p12: f1 channel - 180 degree low power shaped pulse
; choose p12 according to the desired selectivity
;p13: f2 channel - 180 degree low power shaped pulse
; choose p13 according to the desired selectivity
;p32: f1 channel - 180 degree shaped pulse (adiabatic) [20 msec]

;GRADIENT DURATION
;p16: duration of CTP gradients [1 msec]

;DELAY
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(nH*4*J(XH))
;d3 : 1/(nF*4*J(XH))
;d9 : TOCSY mixing time
;d16: delay for gradient recovery [200 usec]
;d30 : illumination time
;d31 : cycle time

;PULSE SHAPE
;spnam11 : f2 channel - file name for the selective 90 exciation shaped pulse [Gauss]
;spnam12 : f1 channel - file name for the selective 180 refocusing shaped pulse [Rsnob]
;spnam13 : f2 channel - file name for the selective 180 refocusing shaped pulse [Rsnob]
;spnam29 : f1 channel - file name for the adiabatic shaped pulse using in ZQF [Crp60,20,20.10]
; smoothed chirp (sweepwidth, 20% smoothing, 10000 points)

;GRADIENT SHAPE
;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam3 : SMSQ10.100
;gpnam4 : SMSQ10.100
;gpnam5 : SMSQ10.100
;gpnam20: SMSQ10.100
;gpnam21: SMSQ10.100

;GRADIENT STRENGTH
;gpz0 : ZQF gradient [3%]
;gpz1 : F-H CTP gradient [47%]
;gpz2 : homospoil gradient [37%]
;gpz3 : 180 1H CTP gradient [41%]
;gpz4 : homospoil gradient TOCSY [17%]
;gpz5 : 180 1H CTP gradient [33%]
;gpz20 : homospoil gradient purge pulses [53%]
;gpz21 : homospoil gradient purge pulses [59%]

;CONSTANTS
;cnst1 : 0 (--) / 1 (Purge pulses)
;cnst2 : J(XH)
;cnst3 : number of protons coupled with fluorine
;cnst7 : number of fluorines coupled with protons
```

;cnst6 : 0 (--) / 1 (ZQF-TOCSY)
;cnst12: chemical shift for selective 1H pulse (offset, in ppm)
;cnst51: Bandwidth of the selective 1H pulse (Hz)
;cnst61 : light duty-cycle (% brightness)

;OTHER
;ns: 2 * n, the total number of scans: NS * TD0
;ds: 2
;set O2 on resonance to the 19F signal of interest
;cpd2: p5m4sp180
;l1: loop for DIPSI cycle: ((p6*115.112) * l1) = mixing time
; if any DELTA become negative try to reduce the gradient duration or selective pulse duration